
Subject: Re: cGImage, Multiplot with /KEEP_ASPECT_RATIO
Posted by [David Fanning](#) on Tue, 12 Feb 2013 13:54:15 GMT
[View Forum Message](#) <> [Reply to Message](#)

Fab writes:

```
> In the following I am making two plots with cg* routines.
>
> img1 = cgDemoData(7)
> img2 = congrid(img1, 360, 180)
> cgDisplay, 900, 400
> !P.Multi = [0, 2, 1]
> multimargin = [1,1,1,1]
> cgImage, img1, MULTIMARGIN=multimargin, /KEEP_ASPECT_RATIO, /AXES
> cgImage, img2, MULTIMARGIN=multimargin, /KEEP_ASPECT_RATIO, /AXES
> !P.Multi = 0
>
> Nice, but I would like the top limit of the plots being the same
> (meaning translate the second plot upwards a bit). I could'nt find an
> easy solution. There must be one though! Any thoughts?
```

POSITION keyword. Heterogeneous positioning of plots is not the forte of !P.MULTI, who is a guy who craves repetition and homogeneity. :-)

And, since you are asking to keep the aspect ratio, you will have to be prepared to adjust the position of the second image in Y space. I would probably try something like this.

```
;-----
img1 = cgDemoData(7)
img2 = congrid(img1, 360, 180)
cgDisplay, 900, 400
cgImage, img1, /KEEP_ASPECT_RATIO, /AXES, $
  POSITION=[0.05, 0.1, 0.475, 0.90], OPOSITION=op
```

```
s = Size(img2, /Dimensions)
imgaspect = Float(s[1])/s[0]
```

```
xdist = 0.95-0.525
ydist = (xdist * imgaspect) * 2
pos2 = [0.525, op[3]-ydist, 0.95, op[3]]
cgImage, img2, /AXES, POSITION=pos2, /NoErase
;-----
```

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.idlcoyote.com/>

Sepore ma de ni thue. ("Perhaps thou speakest truth.")
